

*** ADEQUACY OF PRENATAL CARE UTILIZATION INDEX

Milton Kotelchuck, PhD, MPH

SAS COMPUTATIONAL PROGRAM

Version (3), September, 1994

This SAS program is written in seven major parts. Each section of the program is described and presented separately.

- 1) Variable Definitions
- 2) Data Input
- 3) Acceptable Data Values
 - a) Value ranges
 - b) No prenatal care codes
- 4) Missing Gestational Age Imputation
- 5) Initiation of Prenatal Care Index
- 6) Received Prenatal Care Services Index
 - a) Expected Prenatal Care Visit Calculation
 - b) Observed/Expected Ratio
- 7) Two Factor Summary Index
- 8) Other Statistical Calculations;

```
*-----*
* PART 1: VARIABLE DEFINITIONS *
*-----*
*
* FROM BIRTH CERTIFICATE DATA FILE: *
*
* NPCVBC   Number of prenatal care visits, from birth certificate *
* MPCBBC   Month prenatal care visits began, from birth certificate *
*
* SEXBC    Sex of infant, from birth certificate *
* GAGEBC   Gestational age, from birth certificate *
* BWGRAMS  Birth weight in grams *
```

```

*
*          *
*  CALCULATED WITHIN THIS SAS PROGRAM:
*          *
*  UEXPVIS  Unadjusted expected prenatal care visits
*  EXPVIS   Expected prenatal care visits
*  EVRATIO  Expected visit ratio (observed/expected ratio)
*  EVINDEX  Expected visit index (received PNC service index)
*  MOINDEX4 Month prenatal care initiation index
*  INDEXSUM Two factor summary index
*  NOPNC    No prenatal care received
*          *
*  IMPUTED WITHIN THIS SAS PROGRAM WHEN NECESSARY DATA ARE
MISSING *
*          *
*  GESTIMP  Gestational age imputation marker
*          *
*-----*

```

```

*-----*
* PART 2: DATA INPUT
*-----*

```

```

*
*          *
*  This initial part of the program receives the input data to be
*  analyzed. Each user of this program must adapt this section to
*  his/her own data set and data input files.
*          *
*
*  The critical variables needed to be imputed for this program
*  include: number of prenatal care visits, month of first prenatal
*  care visit, gestational age (date of birth minus date of last
*  menstrual period / 7 days/wk), birth weight (in grams), and sex.
*  Other variables may be added to allow for analysis of adequacy of
*  prenatal care utilization results stratified by those variables.
*          *
*
*  Specifically set the following variables to be equal to your
*  data base's variable names.
*          *
*-----*

```

*** Substitute your equivalent variable name here;

```

NPCVBC = _____;
MPCBBC = _____;
SEXBC  = _____;
GAGEBC = _____;
BWGRAMS = _____;

```

```

*-----*
* PART 3: ACCEPTABLE DATA VALUES
*-----*

```

```

* PART 3a: RANGE EDITS
*-----*
*
*          *
*  This section sets limits on the acceptable values of the
*          *

```

```

* above variables. All unacceptable values are made blank.
* This program assumes all missing and incorrect data are
* blank ( . )
*
*-----*
IF NPCVBC < 0 THEN NPCVBC = .;      *** ACCEPTABLE VALUES 0-90;
IF NPCVBC > 90 THEN NPCVBC = .;

IF MPCBBC < 0 THEN MPCBBC = .;      *** ACCEPTABLE VALUES 0-10;
IF MPCBBC > 10 THEN MPCBBC = .;

IF GAGEBC < 18 OR GAGEBC > 50 THEN GAGEBC = .; *** ACCEPTABLE
VALUES 18-50;

IF (GAGEBC > .) & (MPCBBC > (GAGEBC/4)) THEN MPCBBC=. ;
*** OMTS INCONSISTENT
VALUES OF MPCBBC OR
GAGEBC;

IF SEXBC < 1 or SEXBC > 2 THEN SEXBC = 2; *** ACCEPTABLE VALUES 1,2
UNKNOWN SEX=FEMALE;

IF BWGRAMS < 400 OR BWGRAMS > 6000 THEN BWGRAMS = .; ***
ACCEPTABLE
VALUES 400-
6000;

*-----*
* PART 3b: NO PRENATAL CARE CODES
*-----*
*
* Distinguishing between no prenatal care (0) and missing prenatal
* care (.) is difficult because of known inconsistent coding
* practices used on various state and national data sets. The
* difficulty is most evident in the coding of month prenatal care
* visits began, wherein the code "0" (MPCBBC=0) may have multiple
* meanings. By contrast, the code "0" for number of PNC visits
* (NPCVBC=0) generally means no PNC.
*
* From this revision (3) onward, the default assumption
* of the APNCU Index is that no PNC will be assumed if:
* 1. the number of PNC visits (NPCVBC) = 0 AND month PNC
* began (MPCBBC) = 0 or . (missing), OR
* 2. the month PNC began (MPCBBC) = 0 AND number of visits
* (NPCVBC) = 0 or . (missing)
*
* The default assumptions are written into the program.
* If the default assumptions are not valid for your data set,
* change the SAS code below to accommodate the distinctive coding
* pattern for NO PNC in your data set.
*-----*

```

*** Each of the next 4 questions is followed by a line of SAS code.
Edit each line of SAS code as follows:

If the answer is YES- remove asterisk from SAS code for the item.

If the answer is NO- do not remove asterisk from SAS code for the item.

```
*-----*
* (1) Does the coding MPCBBC = 0 indicate that PNC began in the zero *
* month of PNC (e.g., PNC began before month one)? *
*-----*
*** Edit next line as needed;

* IF MPCBBC = 0 THEN MPCBBC = 1;

*-----*
* (2) Does the coding MPCBBC = 0 indicate that PNC began in month 10 or *
* later? *
*-----*
*** Edit next line as needed;

* IF MPCBBC = 0 THEN MPCBBC = 10;

*-----*
* (3) Does the coding MPCBBC = 0 indicate that data for month PNC began *
* are unknown or missing? *
*-----*
*** Edit next line as needed;

* IF MPCBBC = 0 THEN MPCBBC = . ;

*-----*
* (4) Does the coding NPCVBC = 0 indicate that the number of PNC *
* visits = unknown or missing? *
*-----*
*** Edit next line as needed;

* IF NPCVBC = 0 THEN NPCVBC = . ;

*** End of user edits;

*-----*
* RECODES FOR NO PRENATAL CARE CODES *
*-----*

IF (NPCVBC=0 AND MPCBBC >=1) /* For invalid combinations of */
OR (MPCBBC=0 AND NPCVBC >=1) THEN DO; /* number of PNC visits and */
  NPCVBC=.; /* month PNC began, vars are */
  MPCBBC=.; /* recoded to missing */
END;

IF NPCVBC=0 AND MPCBBC=. THEN MPCBBC=0; /* If number of visits = 0 and */
/* month PNC began is . , month */
```

/* PNC began is recoded to 0 */

IF MPCBBC=0 AND NPCVBC=. THEN NPCVBC=0; /* If month PNC began = 0 and */

/* number of visits is . , */
/* number of visits is recoded */
/* to 0 */

* PART 4: GESTATIONAL AGE EQUIVALENCE *

* *
* Because gestational age is so frequently missing from birth *
* certificates, a gestational age equivalent is imputed from *
* birth weight (only for those records with gestational age *
* missing). *
* *
* The following table shows the sex-linked conversion. A *
* variable called GESTIMP has been created to indicate when *
* a substitution is made (GESTATION IMPUTED 1=NO/2=YES). *
* *
-----;

GESTIMP = .;
IF 18 <= GAGEBC <= 50 THEN GESTIMP = 1; *** (1 = NOT IMPUTED,
2 = IMPUTED);
ELSE GESTIMP = 2;

IF GAGEBC = . THEN DO;
IF SEXBC=1 AND 530 <= BWGRAMS <= 608 THEN GAGEBC=22;
ELSE IF SEXBC=1 AND 609 <= BWGRAMS <= 698 THEN GAGEBC=23;
ELSE IF SEXBC=1 AND 699 <= BWGRAMS <= 799 THEN GAGEBC=24;
ELSE IF SEXBC=1 AND 800 <= BWGRAMS <= 912 THEN GAGEBC=25;
ELSE IF SEXBC=1 AND 913 <= BWGRAMS <= 1040 THEN GAGEBC=26;
ELSE IF SEXBC=1 AND 1041 <= BWGRAMS <= 1183 THEN GAGEBC=27;
ELSE IF SEXBC=1 AND 1184 <= BWGRAMS <= 1342 THEN GAGEBC=28;
ELSE IF SEXBC=1 AND 1343 <= BWGRAMS <= 1536 THEN GAGEBC=29;
ELSE IF SEXBC=1 AND 1537 <= BWGRAMS <= 1751 THEN GAGEBC=30;
ELSE IF SEXBC=1 AND 1752 <= BWGRAMS <= 1978 THEN GAGEBC=31;
ELSE IF SEXBC=1 AND 1979 <= BWGRAMS <= 2219 THEN GAGEBC=32;
ELSE IF SEXBC=1 AND 2220 <= BWGRAMS <= 2458 THEN GAGEBC=33;
ELSE IF SEXBC=1 AND 2459 <= BWGRAMS <= 2693 THEN GAGEBC=34;
ELSE IF SEXBC=1 AND 2694 <= BWGRAMS <= 2909 THEN GAGEBC=35;
ELSE IF SEXBC=1 AND 2910 <= BWGRAMS <= 3111 THEN GAGEBC=36;
ELSE IF SEXBC=1 AND 3112 <= BWGRAMS <= 3291 THEN GAGEBC=37;
ELSE IF SEXBC=1 AND 3292 <= BWGRAMS <= 3433 THEN GAGEBC=38;
ELSE IF SEXBC=1 AND 3434 <= BWGRAMS <= 3533 THEN GAGEBC=39;
ELSE IF SEXBC=1 AND 3534 <= BWGRAMS <= 6000 THEN GAGEBC=40;
ELSE IF SEXBC=2 AND 496 <= BWGRAMS <= 568 THEN GAGEBC=22;
ELSE IF SEXBC=2 AND 569 <= BWGRAMS <= 650 THEN GAGEBC=23;
ELSE IF SEXBC=2 AND 651 <= BWGRAMS <= 744 THEN GAGEBC=24;
ELSE IF SEXBC=2 AND 745 <= BWGRAMS <= 849 THEN GAGEBC=25;
ELSE IF SEXBC=2 AND 850 <= BWGRAMS <= 968 THEN GAGEBC=26;

```

ELSE IF SEXBC=2 AND 969 <= BWGRAMS <= 1101 THEN GAGEBC=27;
ELSE IF SEXBC=2 AND 1102 <= BWGRAMS <= 1251 THEN GAGEBC=28;
ELSE IF SEXBC=2 AND 1252 <= BWGRAMS <= 1429 THEN GAGEBC=29;
ELSE IF SEXBC=2 AND 1430 <= BWGRAMS <= 1636 THEN GAGEBC=30;
ELSE IF SEXBC=2 AND 1637 <= BWGRAMS <= 1860 THEN GAGEBC=31;
ELSE IF SEXBC=2 AND 1861 <= BWGRAMS <= 2089 THEN GAGEBC=32;
ELSE IF SEXBC=2 AND 2090 <= BWGRAMS <= 2328 THEN GAGEBC=33;
ELSE IF SEXBC=2 AND 2329 <= BWGRAMS <= 2561 THEN GAGEBC=34;
ELSE IF SEXBC=2 AND 2562 <= BWGRAMS <= 2787 THEN GAGEBC=35;
ELSE IF SEXBC=2 AND 2788 <= BWGRAMS <= 2991 THEN GAGEBC=36;
ELSE IF SEXBC=2 AND 2992 <= BWGRAMS <= 3160 THEN GAGEBC=37;
ELSE IF SEXBC=2 AND 3161 <= BWGRAMS <= 3293 THEN GAGEBC=38;
ELSE IF SEXBC=2 AND 3294 <= BWGRAMS <= 3388 THEN GAGEBC=39;
ELSE IF SEXBC=2 AND 3389 <= BWGRAMS <= 6000 THEN GAGEBC=40;
END;
IF GAGEBC = . THEN GESTIMP = .;

```

```

*-----*
* PART 5: CALCULATION OF ADEQUACY OF INITIATION OF PRENATAL CARE INDEX *
*-----*
*
* This section calculates the adequacy of initiation of prenatal care. This section is basically a straightforward recoding of month prenatal care began.
*
* Coding: 1=INADEQUATE 2=INTERMEDIATE 3=ADEQUATE 4=ADEQUATE PLUS
* 0=MISSING INFORMATION
*-----*

```

```

MOINDEX4=.;
IF NPCVBC>=0 THEN DO;
  IF 1<= MPCBBC <=2 THEN MOINDEX4=4;
  ELSE IF 3<= MPCBBC <=4 THEN MOINDEX4=3;
  ELSE IF 5<= MPCBBC <=6 THEN MOINDEX4=2;
  ELSE MOINDEX4=1;
END;
IF MPCBBC = . OR NPCVBC = . THEN MOINDEX4=0;

```

```

*-----*
* PART 6: EXPECTED VISITS CALCULATIONS AND RECEIVED PRENATAL CARE SERVICES INDEX *
*-----*
*
* This section calculates the Adequacy of Received Prenatal Care Services Index.
*
* Two principal steps are involved. The first step determines the EXPECTED VISITS for each pregnancy (which requires establishing

```

```

* the recommended visits for the gestational age and then an      *
* adjustment for the timing of the initiation of care). The second *
* step calculates the EXPECTED VISIT RATIO (OBSERVED / EXPECTED)  *
* which is then directly converted to the Received Prenatal Care  *
* Services Index.                                               *
*                                                                 *
*-----*

```

```

*** First step of expected visits calculation adjusts solely
    for length of gestation (based on ACOG recommendations);

```

```

EXPVIS=.; *** Initialize expected & unadjusted expected visits;
UEXPVIS=.;
IF GAGEBC >= 35 THEN UEXPVIS=(GAGEBC-35)+9;
ELSE IF GAGEBC =34 THEN UEXPVIS=9;
ELSE IF GAGEBC >=32 THEN UEXPVIS=8;
ELSE IF GAGEBC >=30 THEN UEXPVIS=7;
ELSE IF GAGEBC >=26 THEN UEXPVIS=6;
ELSE IF GAGEBC >=22 THEN UEXPVIS=5;
ELSE IF GAGEBC >=18 THEN UEXPVIS=4;
ELSE IF GAGEBC >=14 THEN UEXPVIS=3;
ELSE IF GAGEBC >=10 THEN UEXPVIS=2;
ELSE IF GAGEBC >=6 THEN UEXPVIS=1;
ELSE IF GAGEBC >=0 THEN UEXPVIS=0;

```

```

*** Final step of expected visits calculation adjusts for month
    of prenatal care initiation;

```

```

IF MPCBBC = . OR MPCBBC=0 THEN EXPVIS = UEXPVIS;
ELSE IF MPCBBC=10 THEN EXPVIS=UEXPVIS-17;
ELSE IF MPCBBC=9 THEN EXPVIS=UEXPVIS-13;
ELSE IF MPCBBC=8 THEN EXPVIS=UEXPVIS-9;
ELSE IF MPCBBC=7 THEN EXPVIS=UEXPVIS-7;
ELSE IF MPCBBC=6 THEN EXPVIS=UEXPVIS-6;
ELSE IF MPCBBC=5 THEN EXPVIS=UEXPVIS-5;
ELSE IF MPCBBC=4 THEN EXPVIS=UEXPVIS-3;
ELSE IF MPCBBC=3 THEN EXPVIS=UEXPVIS-2;
ELSE IF MPCBBC=2 THEN EXPVIS=UEXPVIS-1;
ELSE IF MPCBBC=1 THEN EXPVIS=UEXPVIS ;
IF EXPVIS <=0 THEN EXPVIS=1;

```

```

*** Calculation of observed / expected ratio (expected visits
    ratio);

```

```

EVRATIO=(NPCVBC/EXPVIS)*100;

```

```

*** Calculation of adequacy of received service (expected visits) index

```

```

    Coding: 1=INADEQUATE 2=INTERMEDIATE 3=ADEQUATE 4=ADEQUATE
    PLUS
    0=MISSING INFORMATION;

```

```

EVINDEX=.;
IF EVRATIO = . THEN EVINDEX=0;
ELSE IF MPCBBC = . THEN EVINDEX=0;
ELSE DO;
  IF EVRATIO > 109.99 THEN EVINDEX=4;
  IF 109.99 >= EVRATIO > 79.99 THEN EVINDEX=3;
  IF 79.99 >= EVRATIO > 49.99 THEN EVINDEX=2;
  IF EVRATIO <= 49.99 THEN EVINDEX=1;
END;

```

```

*-----*
* PART 7: CALCULATION OF SUMMATIVE TWO FACTOR ADEQUACY OF
*
*      PRENATAL CARE UTILIZATION INDEX      *
*-----*
*
*      *
* This section combines the two previously derived factors, *
* Adequacy of Initiation of Prenatal Care Index (MOINDEX4) and *
* Adequacy of Received Prenatal Care Services Index (EVINDEX) *
* into a single summative Adequacy of Prenatal Care Utilization *
* Index. *
*
*      *
* Coding: 1=INADEQUATE 2=INTERMEDIATE 3=ADEQUATE 4=ADEQUATE
PLUS *
*      0=MISSING INFORMATION *
*
*      *
*-----*

```

```

INDEXSUM=.;
IF EVINDEX=0 OR MOINDEX4=0 THEN INDEXSUM=0;
ELSE IF EVINDEX=1 OR 1<=MOINDEX4<=2 THEN INDEXSUM =1;
ELSE IF EVINDEX=3 AND 3<=MOINDEX4<=4 THEN INDEXSUM=3;
ELSE IF EVINDEX=4 AND 3<=MOINDEX4<=4 THEN INDEXSUM=4;
ELSE INDEXSUM =2;

```

```

*-----*
* PART 8: ADDITIONAL STATISTICAL CALCULATIONS WOULD GO IN *
*      THIS SECTION *
*-----*

```

*** Create variable to flag observations with no PNC;

```

IF ((NPCVBC=0) AND (MPCBBC=0 OR MPCBBC=.)
OR ((MPCBBC=0) AND (NPCVBC=0 OR NPCVBC=.)
THEN NOPNC=1; /* 1= no PNC */
ELSE IF NPCVBC=. OR MPCBBC=. THEN NOPNC=.; /* .= missing */
ELSE NOPNC=2; /* 2= some PNC */

```

LABEL

```

NPCVBC= 'Number of prenatal care visits'
MPCBBC= 'Month prenatal care visits began'
SEXBC= 'Sex of infant'

```


GAGEBC= 'Gestational age in weeks'
BWGRAMS= 'Birth weight in grams'
UEXPVIS= 'Unadjusted expected prenatal care visits'
EXPVIS= 'Expected prenatal care visits'
EVRATIO= 'Expected visit ratio (observed/expected)'
EVINDEX= 'Expected visit index (received PNC service)'
MOINDEX4='Month prenatal care initiation index'
INDEXSUM='Two factor summary index'
GESTIMP= 'Gestational age imputation marker'
NOPNC= 'No prenatal care received'
;
run;